
Title stata.com

Intro 3b — New Classical model

Description Remarks and examples Reference Also see

Description
In this example, we solve a New Classical model similar to the one in King and Rebelo (1999). We

also demonstrate how to compare a model’s theoretical predictions under different parameter values
using IRFs.

Remarks and examples stata.com

Remarks are presented under the following headings:
The model
Solving the model
Policy and transition matrices
Impulse responses
Sensitivity analysis

The model
In this model, output, consumption, investment, employment, and other variables are driven by state

variables linked to production and demand. The model is similar to the one in King and Rebelo (1999)
and is referred to as a real business cycle model.

The nonlinear form of the model is

Equation (1) specifies consumption Ct as a function of expected future consumption and the expected
future interest rate Et(Rt+1). Equation (2) specifies labor hours Ht as a function of the wage Wt and
consumption; it is a labor supply equation. Equation (3) is the national income accounting identity for
a closed economy, specifying output Yt as the sum of consumption, investment Xt, and government
spending Gt. Equation (4) is a production function that specifies output as a function of labor input
Ht, capital input Kt, and productivity Zt. Equations (5) and (6) specify labor demand and capital
demand, respectively. Equation (7) specifies the equation for capital accumulation. The model is
completed when we add state transition equations for Zt and Gt. These state transition equations are
conventionally specified after the model has been linearized.

1

http://stata.com
http://stata.com


2 Intro 3b — New Classical model

The linearized form of the model is

ct = Et(ct+1)− (1− β + βδ)Et(rt+1)

ηht = wt − ct

φ1xt = yt − φ2ct − gt

yt = (1− α)(zt + ht) + αkt

wt = yt − ht

rt = yt − kt

kt+1 = δxt + (1− δ)kt

zt+1 = ρzzt + εt+1

gt+1 = ρggt + ξt+1

The model has six control variables and three state variables. Two of the state variables, zt+1 and
gt+1, are modeled as first-order autoregressive processes. The state equation for kt+1 depends on the
current value of a control variable, namely, xt.

Solving the model

The solve option of dsge places the model in state-space form without estimating parameters; it
is similar to iterate(0) but is faster because it does not calculate standard errors. Using solve for
different parameter values of your model is a useful way to explore the model’s theoretical properties.

The parameter values used here are similar to those used in King and Rebelo (1999). Each has
an interpretation. (1 − alpha) is labor’s share of national income. delta is the depreciation rate of
capital. eta is the slope of the labor supply curve. phi1 and phi2 are share parameters related to
investment’s share of national income and consumption’s share of national income, respectively. rhoz
and rhog are autoregressive parameters on the state variables.

. use https://www.stata-press.com/data/r18/usmacro2
(Federal Reserve Economic Data - St. Louis Fed, 2017-01-15)

. dsge (c = F.c - (1-{beta}+{beta}*{delta})*F.r, unobserved)
> ({eta}*h = w - c, unobserved)
> ({phi1}*x = y - {phi2}*c - g, unobserved)
> (y = (1-{alpha})*(z+h) + {alpha}*k)
> (w = y - h, unobserved)
> (r = y - k, unobserved)
> (F.k = {delta}*x+ (1-{delta})*k, state noshock)
> (F.z = {rhoz}*z, state)
> (F.g = {rhog}*g, state),
> from(beta=0.96 eta=1 alpha=0.3 delta=0.025 phi1=0.2 phi2=0.6 rhoz=0.8 rhog=0.3)
> solve noidencheck

DSGE model

Sample: 1955q1 thru 2015q4 Number of obs = 244
Log likelihood = -1957.0261

y Coefficient Std. err. z P>|z| [95% conf. interval]

/structural
beta .96 . . . . .

delta .025 . . . . .
eta 1 . . . . .

phi1 .2 . . . . .
phi2 .6 . . . . .

alpha .3 . . . . .
rhoz .8 . . . . .
rhog .3 . . . . .

sd(e.z) 1 . . .
sd(e.g) 1 . . .

Note: Skipped identification check.
Note: Model solved at specified parameters.

The solve option solves the model at the specified values in from(). We skip the identification
check with noidencheck. Simply solving the model does not involve any reference to the data or
any estimation. Still, we can explore what these parameters imply.

Policy and transition matrices

After solving, we can use many of the postestimation commands, though standard errors will be
missing throughout.

The state transition matrix shows how the state vector in the next period is related to the state
vector in the current period. Some state variables are specified as first-order autoregressive processes,
and their transition equations will simply repeat information that is already available in the estimation
table. However, if any state variable equation contains control variables, then that state variable’s
transition equation will depend on the other state variables.

. estat transition

Transition matrix of state variables

Delta-method
Coefficient std. err. z P>|z| [95% conf. interval]

F.k
k .9256785 . . . . .
z .1078282 . . . . .
g -.1070547 . . . . .

F.z
k 0 (omitted)
z .8 . . . . .
g 0 (omitted)

F.g
k 0 (omitted)
z 0 (omitted)
g .3 . . . . .

Note: Standard errors reported as missing for constrained transition matrix
values.

The value of the state variables z and g in the next period depends only on their value in the
current period, but the value of the capital stock k in the next period depends on the current value
of all three state variables. This feature means that, for example, a shock to the z state variable has
two effects: it increases future values of z, because z is autoregressive, but it also increases future
values of k. Interrelationships among the state variables can generate more interesting patterns in the
IRFs than the AR(1) dynamics that we saw in [DSGE] Intro 3a.

Impulse responses

One way to compare two parameter sets is to graph the impulse response of model variables to a
shock under each parameter set. We first set the impulse–response file with irf set and then add
impulse responses named persistent to the file with irf create.

. irf set rbcirf
(file rbcirf.irf created)
(file rbcirf.irf now active)

. irf create persistent
(file rbcirf.irf updated)

https://www.stata.com/manuals/dsgeintro3a.pdf#dsgeIntro3a

The response of model variables to a shock to z is graphed by typing

. irf graph irf, irf(persistent) impulse(z) response(y c x h w z) noci
> byopts(yrescale)

Step
Graphs by irfname, impulse variable, and response variable

Each graph is labeled with the IRF name, the impulse variable, and the response variable. For
instance, the top-left graph shows the response of consumption to a shock to the zt state variable. The
bottom-right graph shows the response of the state variable zt itself. The state is persistent, which is
not surprising: we set the autoregressive parameter in the zt equation to 0.8.

In the top-left graph, we see that consumption c rises over time before returning to steady state. The
time unit is quarters, so a value of about 0.27, 4 periods after the shock, indicates that consumption
is 0.27% above its steady-state value one year after the shock. Hours worked h are shown in the
top center graph and rise initially before falling below steady state. The real wage w, output y, and
investment x all rise.

Sensitivity analysis

The responses of variables to a shock to z are persistent. Some variables, like consumption and
the wage, show dynamics beyond the simple autoregressive behavior of z itself. To evaluate the role
of persistence in z on the persistence of other model variables, we rerun the dsge command. This
time, we set the persistence of z to a smaller value of 0.6.

. dsge (c = F.c - (1-{beta}+{beta}*{delta})*F.r, unobserved)
> ({eta}*h = w - c, unobserved)
> ({phi1}*x = y - {phi2}*c - g, unobserved)
> (y = (1-{alpha})*(z+h) + {alpha}*k)
> (w = y - h, unobserved)
> (r = y - k, unobserved)
> (F.k = {delta}*x+ (1-{delta})*k, state noshock)
> (F.z = {rhoz}*z, state)
> (F.g = {rhog}*g, state),
> from(beta=0.96 eta=1 alpha=0.3 delta=0.025 phi1=0.2 phi2=0.6 rhoz=0.6
> rhog=0.3) solve noidencheck

DSGE model

Sample: 1955q1 thru 2015q4 Number of obs = 244
Log likelihood = -1659.7331

y Coefficient Std. err. z P>|z| [95% conf. interval]

/structural
beta .96 . . . . .

delta .025 . . . . .
eta 1 . . . . .

phi1 .2 . . . . .
phi2 .6 . . . . .

alpha .3 . . . . .
rhoz .6 . . . . .
rhog .3 . . . . .

sd(e.z) 1 . . .
sd(e.g) 1 . . .

Note: Skipped identification check.
Note: Model solved at specified parameters.

The only change in the parameter set is that rhoz has been set to 0.6 from its earlier setting of
0.8. We can add the impulse responses of this model to the irf file with the name transitory,

. irf create transitory, replace
(irfname transitory not found in rbcirf.irf)
(file rbcirf.irf updated)

and graph them.

. irf graph irf, irf(transitory) impulse(z) response(y c x h w z) noci
> byopts(yrescale)

Step
Graphs by irfname, impulse variable, and response variable

Model variables are much less persistent. We can use irf ograph to overlay the IRF for a variable
under the two calibrations. This way we can view the differences across calibrations directly.

. irf ograph (persistent z c irf) (transitory z c irf)

 persistent: irf of z -> c
 transitory: irf of z -> c

When the shock itself is persistent, consumption responds persistently. When the shock is transitory,
consumption returns to its steady-state value quickly.

Also see
[DSGE] Intro 1 — Introduction to DSGEs

[DSGE] Intro 3a — New Keynesian model

[DSGE] Intro 3c — Financial frictions model

[DSGE] Intro 3e — Nonlinear New Classical model

[DSGE] dsge — Linear dynamic stochastic general equilibrium models

[DSGE] dsge postestimation — Postestimation tools for dsge

